
ABSTRCT 
The Center for Biologics Evaluation and Research (CBER) and European Medicines 
Evaluation Agency (EMEA) have different criteria for calculating immunogenicity fold rise 
when serology data contains Lower Limit of Quantification (LLOQ). This paper introduces 
a multifunctional SAS macro that is capable of calculating the fold rise per both criteria 
with the flexibility to calculate fold rise across different visits and in the meantime 
capturing the LLOQs of multiple antibody strains. 

INTRODUCTION 
In simple sense, the fold rise is calculated as the ratio of the post-vaccination titer value to 
the pre-vaccination value. The geometric mean of the fold rise is called Geometric Mean 
Fold Rise (GMFR). In vaccine studies, however, LLOQ (Lower Limit of Quantification) 
and ULOQ (Upper Limit of Quantification) are very common for antibody titers, especially 
the LLOQ case. The values of LLOQ and ULOQ vary with different antibody strains. In 
practice, the >ULOQ case usually uses the ULOQ as the substitute for the missing value. 
The <LLOQ case, however, is handled differently per different regulator criteria. 
 

Figure 1: A Simple Numeric Illustration of EMEA and CBER criteria 

 
For the calculation of regular Geometric Means of Titers and seroprotection, <LLOQ is 
converted to 0.5*LLOQ per both CBER and EMEA criteria. Things become different when 
it comes to calculate fold rise. When the pre vaccination value is < LLOQ and the post 
value is at or above LLOQ, the CBER criterion converts the < LLOQ to LLOQ but EMEA 

uses 0.5*LLOQ. When the pre vaccination is at or above LLOQ and post vaccination value 
is < LLOQ, both criteria converts the < LLOQ to 0.5*LLOQ. When both the pre and post 
values are < LLOQ, the fold rise is set to 1 per both criteria. In general, the CBER criterion 
is more conservative and more difficult to get higher fold rise values. Figure 1 is a simple 
numeric illustration of the difference between EMEA and CBER criteria. 

 

Figure 2: Example of Fold Rise across Different Visits 

 
 
 
THE FOLD RISE MACRO 
The following SAS macro is developed to calculate fold rise in according to the different 
regulator criteria specified in the previous section. Five macro parameters are created in 
this macro. The dsin parameter specifies the name of the input data set. The testn parameter 
is the total number of strains that need to calculate fold rise. The visit parameter is used to 
list all the visitnumbers involved for fold rise calculation in an ascending order. The sf 
parameter is for suffix. Sometimes in immune-analysis, fold rise needs to be calculated not 
only for the ratio of the last and first blood sample results, but also across other visits, for 
example, the last and the last but one visit. Figure 2 is an example of fold rise across 
different visits. The visit and sf parameter of this macro are designed to deal with such 
situation, which offers the flexibility to calculate fold rise across all the visits listed in the 
visit parameter. 
 

%macro foldrise(dsin=, testn=, visit=); 

proc sql noprint;  

    select distinct paramn into :lloqn separated by ' '  

    from &dsin where index(lborres,'<');  

quit; 

 

%do n=1 %to &testn; 

%if %sysfunc(indexw(&lloqn,&n)) %then %do; 

        proc sql noprint; 

            select distinct compress(lborres,'<') into :lloq&n  

            from &dsin. where paramn=&n & index(lborres,'<');  

        quit; 

    %end; 

    %else %do; 

        %let lloq&n=.; 

    %end; 

%end; 

 

%let num=%sysfunc(countw(&visit)); 

%do v=1 %to %eval(&num-1); 

%let sf=%scan(&visit,&v); 

 

proc sort data=&dsin.;by usubjid paramn visitnum; run; 

 

data base&sf.(rename=(aval=abase&sf.) keep=usubjid paramn aval); 

 set &dsin.(where=(visitnum=&sf.) ); 

proc sort;by usubjid paramn; 

run; 

 

data &dsin; 

 merge &dsin. base&sf.; 

 by usubjid paramn; 

run; 

 

data &dsin.; 

    set &dsin.; 

    %do m=1 %to &testn.; 

        if paramn=&m then lloq&m.=&&lloq&m.; 

        if visitnum>&sf. & paramn=&m. then do; 

            if .<abase&sf.<&&lloq&m. & .< aval <&&lloq&m. then do; 

                IMFOLDCO&sf.=1; IMFOLD&sf.=1;  

            end; 

            if .<abase&sf.<&&lloq&m. & aval>=&&lloq&m. then do; 

    IMFOLDCO&sf.= aval /&&lloq&m.;  

    IMFOLD&sf.= aval /abase&sf.; 

            end; 

            if abase&sf.>=&&lloq&m. & aval >=&&lloq&m. then do; 

    IMFOLDCO&sf.= aval /abase&sf.;  

    IMFOLD&sf.= aval /abase&sf.; 

            end; 

            if abase&sf.>=&&lloq&m. & .< aval <&&lloq&m. then do; 

    IMFOLDCO&sf.=0.5*&&lloq&m/abase&sf.;  

    IMFOLD&sf. = aval /abase&sf.; 

            end; 

        end; 

    %end;   

run; 

%end; 

%mend; 
 

IMFOLD and IMFOLDCO variables are fold rise by EMEA and CBER criteria 
respectively. The value of LLOQ for each test varies. At the beginning of this macro, the 
LLOQs of each strain are collected and stored in a string. For those serology tests that don’t 
have a LLOQ, its LLOQ value is set to null. Capturing these LLOQs is very useful because 
they are also used for seroprotection and booster response rate calculation.  
 
 
CONCLUSION 
Fold rise is a key statistics of immunogenicity analysis in vaccine clinical studies. The SAS 
macro introduced in this paper simultaneously calculates the fold rise in according to both 
EMEA and CBER criteria. In the meantime, this macro can calculate fold rise for all the 
visits listed in the macro parameter, which in turn offers the flexibility to calculate fold rise 
across different visits. In addition, this macro can also capture the LLOQs of multiple 
antibody strains, thus facilitate vaccine clinical programming. 

